

CSAPP book is **very useful** and well-aligned with class for the remainder of the course.

## C to Machine Code and x86 Basics

ISA context and x86 history

Translation tools: C --> assembly <--> machine code

x86 Basics:

- Registers
- Data movement instructions
- Memory addressing modes
- Arithmetic instructions


2

## Turning C into Machine Code

### C Code

```
void sumstore(long x, long y,
              long *dest) {
    long t = x + y;
    *dest = t;
}
```

sum.c




gcc -Og -S sum.c

### Generated x86 Assembly Code

Human-readable language close to machine code.

```
sum:
    addq %rdi,%rsi
    movq %rsi,(%rdx)
    retq
```

sum.s




### Object Code

```
01010101100010011110010110
00101101000101000011000000
00110100010100001000100010
01111011000101110111000011
```

sum.o

### Executable: sum

Resolve references between object files,  
libraries, (re)locate data



linker

3

## Disassembling Object Code

Disassembled by `objdump -d sum`

```
000000000400536 <sumstore>:
400536: 48 01 fe add %rdi,%rsi
400539: 48 89 32 mov %rsi,(%rdx)
40053c: c3      retq
```

Disassembler

```
001011010001010000110000
001101000101000010001000
011110110001011101110000
...
```

### Object

```
0x00400536:
0x48
0x01
0xfe
0x48
0x89
0x32
0xc3
```

### Disassembled by GDB

```
0x000000000400536 <+0>: add %rdi,%rsi
0x000000000400539 <+3>: mov %rsi,(%rdx)
0x00000000040053c <+6>: retq
```

\$ gdb sum

(gdb) disassemble sumstore

(disassemble function)

(gdb) x/7b sum

(examine the 13 bytes starting at sum)

5


## x86-64 registers

%rax	Return Value
%rbx	
%rcx	Argument 4
%rdx	Argument 3
%rsi	Argument 2
%rdi	Argument 1
%rsp	Special Purpose: Stack Pointer
%rbp	
%r8	Argument 5
%r9	Argument 6
%r10	
%r11	
%r12	
%r13	
%r14	
%r15	

64-bits / 8 bytes

### sub-registers


1985: 32-bit extended register %eax  
1978: 16-bit register %ax



high and low bytes of %ax

Low 32 bits of %rsi      Low 16 bits of %rsi

### historical artifacts



32-bit sub-register to match

Some have special uses for particular instructions

# x86: Three Basic Kinds of Instructions

## 1. Data movement between memory and register

**Load** data from memory into register

$\%reg \leftarrow Mem[address]$

**Store** register data into memory

$Mem[address] \leftarrow \%reg$

Memory is an array[] of bytes!

## 2. Arithmetic/logic on register or memory data

$c = a + b;$                        $z = x \ll y;$                        $i = h \& g;$

## 3. Comparisons and Control flow to choose next instruction

Unconditional jumps to/from procedures

Conditional branches

# Data movement instructions

**mov** Source, Dest

data size    is one of {**b**, **w**, **l**, **q**}

movq: move 8-byte "quad word"

movl: move 4-byte "long word"

movw: move 2-byte "word"

movb: move 1-byte "byte"

Historical terms based on the 16-bit days, not the current machine word size (64 bits)

## Source/Dest operand types:

**Immediate:** Literal integer data

Examples:            \$0x400                      \$-533

**Register:** One of 16 registers

Examples:            %rax                      %rdx

**Memory:** consecutive bytes in memory, at address held by register

Direct addressing:            (%rax)

With displacement/offset:    8(%rsp)

# Memory Addressing Modes

**Indirect**                      (R)                      Mem[Reg[R]]

Register R specifies memory address: **movq (%rcx), %rax**

**Displacement**            D(R)                      Mem[Reg[R]+D]

Register R specifies **base** memory address (e.g. base of an object)

**Displacement D** specifies literal **offset** (e.g. a field in the object)

**movq %rdx, 8(%rsp)**

General Form: **D(Rb,Ri,S)** Mem[Reg[Rb] + S\*Reg[Ri] + D]

D:        Literal "displacement" value represented in 1, 2, or 4 bytes

Rb:        Base register: Any register

Ri:        Index register: Any except %rsp


S:        Scale: 1, 2, 4, or 8

# Pointers and Memory Addressing

```
void swap(long* xp, long* yp){
    long t0 = *xp;
    long t1 = *yp;
    *xp = t1;
    *yp = t0;
}
```

```
swap:
    movq (%rdi), %rax
    movq (%rsi), %rdx
    movq %rdx, (%rdi)
    movq %rax, (%rsi)
    retq
```

Register	Variable
%rdi	↔ xp
%rsi	↔ yp
%rax	↔ t0
%rdx	↔ t1



# Address Computation Examples



## General Addressing Modes

$D(Rb, Ri, S)$	$Mem[Reg[Rb] + S * Reg[Ri] + D]$	
<b>Special Cases:</b>		<b>Implicitly:</b>
$(Rb, Ri)$	$Mem[Reg[Rb] + Reg[Ri]]$	$(S=1, D=0)$
$D(Rb, Ri)$	$Mem[Reg[Rb] + Reg[Ri] + D]$	$(S=1)$
$(Rb, Ri, S)$	$Mem[Reg[Rb] + S * Reg[Ri]]$	$(D=0)$

### Register contents

<code>%rdx</code>	<code>0xf000</code>
<code>%rcx</code>	<code>0x100</code>

Address Expression	Address Computation	Address
<code>0x8(%rdx)</code>		
<code>(%rdx,%rcx)</code>		
<code>(%rdx,%rcx,4)</code>		
<code>0x80(,%rdx,2)</code>		

Compute address given by this addressing mode expression and store it here.

# Load effective address

`leaq Src, Dest`

**DOES NOT ACCESS MEMORY**



**Uses:** "address of" "Lovely Efficient Arithmetic"  
 $p = \&x[i];$   $x + k * I,$  where  $k = 1, 2, 4,$  or  $8$

## leaq vs. movq


Registers	Memory	Address
<code>%rax</code>	<code>0x400</code>	<code>0x120</code>
<code>%rbx</code>	<code>0xf</code>	<code>0x118</code>
<code>%rcx</code>	<code>0x8</code>	<code>0x110</code>
<code>%rdx</code>	<code>0x100</code>	<code>0x108</code>
<code>%rdi</code>		
<code>%rsi</code>	<code>0x1</code>	<code>0x100</code>

```
leaq (%rdx,%rcx,4), %rax
movq (%rdx,%rcx,4), %rbx
leaq (%rdx), %rdi
movq (%rdx), %rsi
```

# Call Stack

Memory region for temporary storage managed with stack discipline.


`%rsp` holds lowest stack address (address of "top" element)



# Call Stack: Push, Pop


## pushq Src

1. Fetch value from `Src`
2. Decrement `%rsp` by 8 (why 8?)
3. Store value at new address given by `%rsp`



## popq Dest

1. Load value from address `%rsp`
2. Write value to `Dest`
3. Increment `%rsp` by 8



Those bits are still there; we're just not using them.

## Procedure Preview (more soon)

*call, ret, push, pop*

Procedure arguments passed in 6 registers:


%rax	Return Value	%r8	Argument 5
%rbx		%r9	Argument 6
%rcx	Argument 4	%r10	
%rdx	Argument 3	%r11	
%rsi	Argument 2	%r12	
%rdi	Argument 1	%r13	
%rsp	Stack pointer	%r14	
%rbp		%r15	

Return value in %rax.

Allocate/push new *stack frame* for each procedure call.

Some local variables,  
saved register values, extra arguments

Deallocate/pop frame before return.



22

## Arithmetic Operations

Two-operand instructions:

**Format**

**addq** *Src, Dest*  
**subq** *Src, Dest*  
**imulq** *Src, Dest*  
**shlq** *Src, Dest*  
**sarq** *Src, Dest*  
**shrq** *Src, Dest*  
**xorq** *Src, Dest*  
**andq** *Src, Dest*  
**orq** *Src, Dest*

**Computation**

$Dest = Dest + Src$   
 $Dest = Dest - Src$   
 $Dest = Dest * Src$   
 $Dest = Dest \ll Src$   
 $Dest = Dest \gg Src$   
 $Dest = Dest \gg Src$   
 $Dest = Dest \wedge Src$   
 $Dest = Dest \& Src$   
 $Dest = Dest | Src$

← **argument order**

*a.k.a salq*  
**Arithmetic**  
**Logical**

One-operand (unary) instructions

**incq** *Dest*       $Dest = Dest + 1$   
**decq** *Dest*       $Dest = Dest - 1$   
**negq** *Dest*       $Dest = -Dest$   
**notq** *Dest*       $Dest = \sim Dest$

increment  
 decrement  
 negate  
 bitwise complement

See CSAPP 3.5.5 for: *mulq, cqto, idivq, divq*


23

## leaq for arithmetic

```
long arith(long x, long y,
           long z){
    long t1 = x+y;
    long t2 = z+t1;
    long t3 = x+4;
    long t4 = y * 48;
    long t5 = t3 + t4;
    long rval = t2 * t5;
    return rval;
}
```

Register	Use(s)
%rdi	Argument <b>x</b>
%rsi	Argument <b>y</b>
%rdx	Argument <b>z</b>
%rax	
%rcx	

```
arith:
    leaq    (%rdi,%rsi), %rax
    addq   %rdx, %rax
    leaq   (%rsi,%rsi,2), %rdx
    salq   $4, %rdx
    leaq   4(%rdi,%rdx), %rcx
    imulq  %rcx, %rax
    ret
```



## Another example

```
long logical(long x, long y){
    long t1 = x^y;
    long t2 = t1 >> 17;
    long mask = (1<<13) - 7;
    long rval = t2 & mask;
    return rval;
}
```

Register	Use(s)
%rdi	Argument <b>x</b>
%rsi	Argument <b>y</b>
%rax	

```
logical:
    movq   %rdi, %rax
    xorq   %rsi, %rax
    sarq   $17, %rax
    andq   $8185, %rax
    retq
```

25